

Luftdruck und -dichte als Funktion der Höhe

Version 2024-12-18

Copyright und Urheberrecht

Dieses Dokument wurde von
Gustav Grönnäs
See: <https://www.gustav-gronnas.de/books/books.html>
mit Euler Math Toolbox
See: <http://euler.rene-grothmann.de>
erstellt,
es darf, sofern es nicht verändert wird, frei und unter Angabe der Quelle, verwendet werden.

Anwendungsgebiete

Dieses Dokument wurde für Studierende im Ingenieurwesen (insbesondere Flugzeugbau und verwandte Bereiche) erstellt. Es dient der Beschreibung der Verfahren, mit denen Flughöhen oder auch -geschwindigkeiten mittels des Luftdruckes oder der Luftdichte gemessen werden. Dabei wird die Unsicherheit der Messung über die verwendeten Gleichungen deutlich.
Die thermodynamischen Grundlagen (etwa bezüglich des Temperaturgradienten, etc.) werden nicht betrachtet. Für Beschreibungen der Atmosphärenphysik, der Meteorologie, etc. sind die hier angestellten Betrachtungen nicht hinreichend.

Sofern nicht anders angegeben, sind alle Größen in den Grundeinheiten angegeben

Die von 'normalen' Flugzeugen genutzte untere Atmosphärenschicht, in der auch Wetterphänomene auftreten, die Troposphäre, erstreckt sich bis in eine Höhe von 18 km am Äquator und 8 km am Pol. Die nachfolgende Beschreibung des atmosphärischen Druckes p bezieht sich ausschließlich auf diesen Bereich. In der Literatur wird häufig eine Begrenzung auf 11 km Höhe angegeben.

Vereinfacht werde die maximal betrachtete Höhe - also nicht notwendigerweise der Gültigkeitsbereich der Gleichungen - zu

$>h_{\max} := 20000;$

gesetzt.

Der Druck p einer Säule auf ihre Grundfläche A ergibt sich aus ihrer Gewichtskraft (als Produkt aus Masse m und Erdbeschleunigung g) geteilt durch ihre Grundfläche

$$p = \frac{mg}{A}$$

Der Druck nimmt also mit der Masse zu, so dass für veränderliche Massen auch der Druck variabel beschrieben sein muss. Es gilt also für differentiell kleine Veränderungen

$$dp = \frac{g}{A} dm$$

Die Masse ist mittels der Gas-Molekülmasse M und der Stoffmenge n beschreibbar

$$m = nM, \text{ mit } M_{Luft} = 28.91 \cdot 10^{-3} \frac{kg}{mol}$$

$$dp = \frac{Mg}{A} dn$$

Aus dem universellen (und idealisierten) Gasgesetz

$$\frac{pV}{\theta} = nR_m, \text{ mit } R_m = 8.31441 \frac{J}{molK}$$

kann die Stoffmenge im betrachteten Volumen V beschrieben

$$\frac{pV}{\theta R_m} = n$$

und ersetzt werden

$$dp = \frac{pMg}{A\theta R_m} dV$$

Das Volumen lässt sich hier gegen die Grundfläche kürzen, so dass nur die Höhe h der Säule in der Gleichung verbleibt

$$dp = \frac{pMg}{\theta R_m} dh$$

Sortiert nach konstanten und variablen Ausdrücken gilt also

$$(1) : dp = p \frac{Mg}{R_m \theta} dh$$

Die Lufttemperatur nimmt (mit guter Näherung für feuchte Luft) mit der Höhe linear ab

$$a = 0.0065 \frac{K}{m}$$

Daher kann die obige Differentialgleichung des Druckes noch um eine Temperaturkompensation verbessert werden

$$dp = p \frac{Mg}{R_m(\theta - ah)} dh$$

Diese Gleichung lässt sich in ihrer Struktur etwas verändern, so dass in nachfolgenden Rechnungen einheitenlose Potenzen erscheinen:

$$(2) : dp = p \frac{Mg}{R_m \theta} \frac{1}{(1 - \frac{a}{\theta} h)} dh$$

Beide Differentialgleichungen lassen sich mittels 'Trennung der Variablen' lösen. Der Leser oder die Leserin möge dieses selbst nachrechnen. Die Lösungen finden sich nachfolgend als 'internationale Höhenformel' und 'exponentielle Formel' verwendet.

Als Standardtemperatur am Boden wird

$$\theta = 15^\circ C = 288K$$

verwendet.

```
>theta0 := 288;
```

Anmerkung: In Folge der Erdrotation ist der Erdbeschleunigung die Zentralbeschleunigung überlagert. Die hieraus resultierende Abweichung der effektiven Erdbeschleunigung vom Normwert liegt unter 1% (der Leser oder die Leserin prüfe dieses selber nach). Da aber insbesondere das Wettergeschehen einen deutlich größeren Einfluss auf den Luftdruck hat, sei die Erdbeschleunigung hier als konstant angenommen.

Bei konstanter Temperatur ist die Luftdichte

$$\rho = \frac{m}{V}$$

proportional zum Luftdruck, denn mit dem universellen Gasgesetz (hier in der Umstellung)

$$V = \frac{n R_m \theta}{p}$$

folgt

$$\rho = \frac{m p}{n R_m \theta}$$

$$\rho = \frac{M}{R_m \theta} p$$

Da aber

$$\theta = \theta_0 - a h$$

gilt, folgt die geringfügig geringere Abnahme der Dichte mit der Höhe, verglichen mit der Abnahme des Druckes

$$(3) : \rho(h) = \frac{M}{R_m} \frac{1}{\theta_0 - a h} p(h)$$

Für eine Geschwindigkeitsmessung wird zumeist der Staudruck einer Strömung

$$p_{stau} = \frac{1}{2} \rho v^2$$

verwendet.

Der gemessene Staudruck kann für die Dichten am Boden mit der realen Geschwindigkeit und in der Höhe beschrieben werden. Diese Gleichungen in einander eingesetzt, führen auf eine Gleichung der gemessenen Geschwindigkeit in der Höhe

$$(4) v_{mess} = \sqrt{\frac{\rho(h)}{\rho_0}} v_0$$

Sei nachfolgend noch die Beispielgeschwindigkeit gewählt:

`>v0 := 100;`

Erdbeschleunigung

```
>g := 9.80665; ...
```

Molekülmasse der Luft

```
>MLuft := 28.91e-3; ...
```

Universelle Gaskonstante

```
>Rm := 8.31441; ...
```

Norm-Luftdruck am Boden [k Pa]

```
>p0 := 101.3; ...
```

Norm-Luftdichte am Boden für 15°C

```
>rho0 = 1.223; ...
```

Standard-Temperatur am Boden

```
>theta0 := 288; ...
```

Temperaturgradient

```
>a := 0.0065; ...
```

Hieraus folgen die Koeffizienten der Luftdruckfunktionen

```
>kt := MLuft *g /(Rm *theta0), ...  
>1/kt |" m", ...  
>ka := MLuft *g /( Rm *a)
```

```
0.000118398129932  
8446.0793475 m  
5.24594791082
```

Anmerkung:

In den nachfolgenden Gleichungen werden die Koeffizienten der Literaturangaben (nicht die, der obigen Berechnungen) verwendet.

internationale Höhenformel

(KUCHLING, H.: Physik - Formeln und Gesetze; VEB Fachbuchverlag Leipzig; 1984)

$$p = 101.3 \text{ kPa} \left(1 - \frac{0.0065}{288} h \right)^{5.255}$$

```
>function f1(x) := p0 *(1 - 0.0065 *x /288)^5.255;
```

exponentielle Formel

(Wikipedia)

$$p = 101.3 \text{ kPa} e^{\left(-\frac{h}{8435 \text{ m}}\right)}$$

```
>function f2(x) := p0 *exp(-x /8435);
```

Hyperbelformel

(RWTH Aachen)

$$p = 101.3 \text{ kPa} \frac{20000m - h}{20000m + h}$$

```
>function f3(x) := p0 *(20000 -x) /(20000 +x);
```

internationale Höhenformel

$$h = \frac{288}{0.0065} m \left(1 - \left(\frac{p}{101.3 \text{ kPa}} \right)^{\frac{1}{5.255}} \right), p \text{ in [kPa]}$$

```
>function fi1(x) := 288 / 0.0065 * (1 - (x / p0)^(1/5.255) );
```

exponentielle Formel

$$h = 8435 m \ln \left(\frac{101.3 \text{ kPa}}{p} \right), p \text{ in [kPa]}$$

```
>function fi2(x) := 8435 * ln(p0 / x);
```

Hyperbelformel

$$h = 20000 m \frac{101.3 \text{ kPa} - p}{101.3 \text{ kPa} + p}, p \text{ in [kPa]}$$

```
>function fi3(x) := 20000 * (p0 - x) / (p0 + x);
```

internationale Höhenformel
(incl. Temperatureinfluss)

$$\rho = 1.223 \frac{kg}{m^3} \left(1 - \frac{0.0065}{288} h \right)^{5.255-1}$$

```
>function fd1(x) := rho0 *(1 - 0.0065 *x /288)^4.255;
```

exponentielle Formel
(ohne Temperatureinfluss)

$$\rho = 1.223 \frac{kg}{m^3} e^{\left(-\frac{h}{8435 m}\right)}$$

```
>function fd2(x) := rho0 *exp(-x /8435);
```

Hyperbelformel
(ohne theoretischen Hintergrund)

$$\rho = 1.223 \frac{kg}{m^3} \frac{20000m - h}{20000m + h}$$

```
>function fd3(x) := rho0 *(20000 -x) /(20000 +x);
```

Relative Abweichung der exponentiellen Druckfunktion von der internationalen Höhenformel

$$\Delta p_{rel} = 100\% \frac{p_{exp} - p_{int}}{p_{int}}$$

```
>function rf2(x) := 100 *(f2(x) -f1(x)) /(f1(x));
```

Relative Abweichung der hyperbolischen Druckfunktion von der internationalen Höhenformel

$$\Delta p_{rel} = 100\% \frac{p_{hyp} - p_{int}}{p_{int}}$$

```
>function rf3(x) := 100 *(f3(x) -f1(x)) /(f1(x));
```

Abweichung der exponentiellen Höhenfunktion von der internationalen Höhenformel

$$\Delta h = h_{exp} - h_{int}$$

```
>function afi2(x) := fi2(x) -fi1(x);
```

Abweichung der hyperbolischen Höhenfunktion von der internationalen Höhenformel

$$\Delta h = h_{hyp} - h_{int}$$

```
>function afi3(x) := fi3(x) -fi1(x);
```

Abweichung der exponentiellen Höhenfunktion von der internationalen Höhenformel abhängig von der Höhe

$$\Delta h = h_{exp} - h$$

```
>function aafi2(x) := fi2(f1(x)) -x;
```

Abweichung der hyperbolischen Höhenfunktion von der internationalen Höhenformel abhängig von der Höhe

$$\Delta h = h_{hyp} - h$$

```
>function aafi3(x) := fi3(f1(x)) -x;
```

Abweichung der exponentiellen Dichtefunktion von der internationalen Dichteformel abhängig von der Höhe

$$\Delta\rho = \frac{\rho_{exp} - \rho_{int}}{\rho_{int}}$$

```
>function aafd2(x) := 100 *(fd2(x) -fd1(x)) /fd1(x);
```

Abweichung der hyperbolischen Dichtefunktion von der internationalen Dichteformel abhängig von der Höhe

$$\Delta\rho = \frac{\rho_{hyp} - \rho_{int}}{\rho_{int}}$$

```
>function aafd3(x) := 100 *(fd3(x) -fd1(x)) /fd1(x);
```

Geschwindigkeitsfunktion aus der internationalen Dichtefunktion abhängig von der Höhe

$$v = \sqrt{\frac{\rho_{int}}{\rho_0}} v_0$$

```
>function fv1(x) := sqrt( fd1(x) /rho0 ) *100;
```

Geschwindigkeitsfunktion aus der exponentiellen Dichtefunktion abhängig von der Höhe

$$v = \sqrt{\frac{\rho_{exp}}{\rho_0}} v_0$$

```
>function fv2(x) := sqrt( fd2(x) /rho0 ) *100;
```

Geschwindigkeitsfunktion aus der hyperbolischen Dichtefunktion abhängig von der Höhe

$$v = \sqrt{\frac{\rho_{hyp}}{\rho_0}} v_0$$

```
>function fv3(x) := sqrt( fd3(x) /rho0 ) *100;
```

Relativer Geschwindigkeitsmessfehler der internationalen Dichtefunktion abhängig von der Höhe

$$\Delta v = 100\% \frac{v_{int} - v_0}{v_0}$$

```
>function aafv1(x) := 100 * (fv1(x) -v0) /v0;
```

Relativer Geschwindigkeitsmessfehler der exponentiellen Dichtefunktion abhängig von der Höhe

$$\Delta v = 100\% \frac{v_{exp} - v_0}{v_0}$$


```
>function aafv2(x) := 100 * (fv2(x) -v0) /v0;
```

Relativer Geschwindigkeitsmessfehler der hyperbolischen Dichtefunktion abhängig von der Höhe

$$\Delta v = 100\% \frac{v_{hyp} - v_0}{v_0}$$

```
>function aafv3(x) := 100 * (fv3(x) -v0) /v0;
```


```
>aspect(1.5); setfont(12pt, 12cm); ...
>plot2d("f1", a=0, b=hmax, c=0, d=110, color=green, <frame); ...
>plot2d("f2", a=0, b=hmax, c=0, d=110, >add, color=blue); ...
>plot2d("f3", a=0, b=hmax, c=0, d=110, >add, color=gray, style="-"); ...
>labelbox(["Hyperbel", "exponentiell", "internat."], ...
>styles=["-", "-", "-"], ...
>colors=[gray, blue, green], ...
>tcolor=[gray, blue, green], ...
>color=transparent, ...
>x=0.95, y=0.15, w=0.45, wt=0.6 ); ...
>title("Luftdruck p = f( h )"); ...
>xlabel("h [m]"); ...
>ylabel("p [k Pa]"); ...
>insimg(); ...
>aspect();
```




```

>aspect(1.5); setfont(12pt, 12cm); ...
>plot2d("fi1", a=0, b=110, c=0, d=hmax, color=green, <frame); ...
>plot2d("fi2", a=0, b=110, c=0, d=hmax, >add, color=blue); ...
>plot2d("fi3", a=0, b=110, c=0, d=hmax, >add, color=gray, style="-"); ...
>labelbox(["Hyperbel", "exponentiell", "internat."], ...
>styles=["-", "-", "-"], ...
>colors=[gray, blue, green], ...
>tcolor=[gray, blue, green], ...
>color=transparent, ...
>x=0.95, y=0.15, w=0.45, wt=0.6 ); ...
>title("Höhe h = f( p )"); ...
>xlabel("p [k Pa]"); ...
>ylabel("h [m]"); ...
>insimg(); ...
>aspect();


```



```
>aspect(1.5); setfont(12pt, 12cm); ...
>plot2d("fd1", a=0, b=hmax, c=0, d=1.4, color=green, <frame); ...
>plot2d("fd2", a=0, b=hmax, c=0, d=1.4, >add, color=blue); ...
>plot2d("fd3", a=0, b=hmax, c=0, d=1.4, >add, color=gray, style="-"); ...
>labelbox(["Hyperbel", "exponentiell", "internat."], ...
>styles=["-", "-", "-"], ...
>colors=[gray, blue, green], ...
>tcolor=[gray, blue, green], ...
>color=transparent, ...
>x=0.95, y=0.15, w=0.45, wt=0.6 ); ...
>title(u"Luftdichte &rho; = f( h )"); ...
>xlabel("h [m]"); ...
>ylabel(u"&rho; [kg / m^3]"); ...
>insimg(); ...
>aspect();
```




```
>aspect(1.5); setfont(12pt, 12cm); ...
>plot2d("fv1", a=0, b=hmax, c=0, d=100, color=green, <frame); ...
>plot2d("fv2", a=0, b=hmax, c=0, d=100, >add, color=blue); ...
>plot2d("fv3", a=0, b=hmax, c=0, d=100, >add, color=gray, style="-"); ...
>labelbox(["Hyperbel", "exponentiell", "internat."], ...
>styles=["-", "-", "-"], ...
>colors=[gray, blue, green], ...
>tcolor=[gray, blue, green], ...
>color=transparent, ...
>x=0.95, y=0.15, w=0.45, wt=0.6 ); ...
>title("Geschwindigkeitsmessung v = f( h )"); ...
>xlabel("h [m]"); ...
>ylabel("v [%]"); ...
>insimg(); ...
>aspect();
```



```


>aspect(1.5); setfont(12pt, 12cm); ...
>plot2d("rf2", a=0, b=hmax, c=-100, d=100, color=blue, <frame); ...
>plot2d("rf3", a=0, b=hmax, c=-100, d=100, >add, color=gray, style="-"); ...
>labelbox(["Hyperbel", "exponentiell"], ...
>styles=["-", "-"], ...
>colors=[gray, blue], ...
>tcolor=[gray, blue], ...
>color=transparent, ...
>x=0.7, y=0.05, w=0.45, wt=0.6 ); ...
>title(u"rel. Druckabweichung  $\Delta p_{rel} = f(h)$ "); ...
>xlabel("h [m]"); ...
>ylabel(u"<math>\Delta p_{rel}</math> [%]"); ...
>insimg(); ...
>aspect();

```




Genauigkeits-Graphen der Höhe (abhängig vom Druck)

```
>aspect(1.5); setfont(12pt, 12cm); ...  
>plot2d("afi2", a=0, b=100, c=-1000, d=4000, color=blue, <frame); ...  
>plot2d("afi3", a=0, b=100, c=-1000, d=4000, >add, color=gray, style="-"); ...  
>labelbox(["Hyperbel", "exponentiell"], ...  
>styles=["-", "-"], ...  
>colors=[gray, blue], ...  
>tcolor=[gray, blue], ...  
>color=transparent, ...  
>x=0.70, y=0.05, w=0.45, wt=0.6 ); ...  
>title(u"abs. Höhenabweichung  $\Delta h = f(p)$ "); ...  
>xlabel("p [k Pa]"); ...  
>ylabel(u"<math>\Delta h</math> [m]"); ...  
>insimg(); ...  
>aspect();
```



Genauigkeits-Graphen der Höhe (abhängig von der Höhe)


```
>aspect(1.5); setfont(12pt, 12cm); ...  
>plot2d("aafi2", a=0, b=hmax, c=-1000, d=4000, color=blue, <frame); ...  
>plot2d("aafi3", a=0, b=hmax, c=-1000, d=4000, >add, color=gray, style="-"); ...  
>labelbox(["Hyperbel", "exponentiell"], ...  
>styles=["-", "-"], ...  
>colors=[gray, blue], ...  
>tcolor=[gray, blue], ...  
>color=transparent, ...  
>x=0.7, y=0.05, w=0.45, wt=0.6 ); ...  
>title(u"abs. Höhenabweichung  $\Delta h = f(h)$ "); ...  
>xlabel("h [m]"); ...  
>ylabel(u"<math>\Delta h</math> [m]"); ...  
>insimg(); ...  
>aspect();
```




```

>aspect(1.5); setfont(12pt, 12cm); ...
>plot2d("aafd2", a=0, b=hmax, c=-20, d=20, color=blue, <frame); ...
>plot2d("aafd3", a=0, b=hmax, c=-20, d=20, >add, color=gray, style="-"); ...
>labelbox(["Hyperbel", "exponentiell"], ...
>styles=["-", "-"], ...
>colors=[gray, blue], ...
>tcolor=[gray, blue], ...
>color=transparent, ...
>x=0.7, y=0.05, w=0.45, wt=0.6 ); ...
>title(u"rel. Dichteabweichung  $\Delta\rho_{rel} = f(h)$ "); ...
>xlabel("h [m]"); ...
>ylabel(u"<math>\Delta\rho_{rel}</math> [%]"); ...
>insimg(); ...
>aspect();

```



```
>aspect(1.5); setfont(12pt, 12cm); ...  
>plot2d("aafv1", a=0, b=hmax, c=-100, d=0, color=green, <frame); ...  
>plot2d("aafv2", a=0, b=hmax, c=-100, d=0, >add, color=blue, <frame); ...  
>plot2d("aafv3", a=0, b=hmax, c=-100, d=0, >add, color=gray, style="-"); ...  
>labelbox(["Hyperbel", "exponentiell", "international"], ...  
>styles=["-", "-", "-"], ...  
>colors=[gray, blue, green], ...  
>tcolor=[gray, blue, green], ...  
>color=transparent, ...  
>x=0.85, y=0.05, w=0.45, wt=0.6 ); ...  
>title(u"rel. Geschwindigkeitsabweichung  $\Delta v_{rel} = f(h)$ "); ...  
>xlabel("h [m]"); ...  
>ylabel(u"<math>\Delta v_{rel}</math> [%]"); ...  
>insimg(); ...  
>aspect();
```



```
>h:=10000; ...
```

Luftdruck

```
>f1(h), ...  
>f2(h), ...  
>f3(h)
```

```
26.4145962907  
30.9554773  
33.7666666667
```

Abweichung von der int. Höhenformel

```
>Dp2 = rf2(h) | " %", ...  
>Dp3 = rf3(h) | " %"
```

```
17.1908022343 %  
27.833362642 %
```

Höhe abhängig vom Druck

```
>p := 100; ...  
>fi1(p), ...  
>fi2(p), ...  
>fi3(p)
```

```
108.769819103  
108.948360123  
129.160457029
```

Höhenfehler

```
>Dh2 = afi2(p), ...  
>Dh3 = afi3(p)
```

```
0.17854102012  
20.3906379261
```

Höhenfehler

```
>h := 1000;  
>DDh2 = aafi2(h), ...  
>DDh3 = aafi3(h)
```

```
11.8736589837  
198.176570429
```

Dichte

```
>rho1 = fd1(h), ...  
>rho2 = fd2(h), ...  
>rho3 = fd3(h)
```

```
1.10979307406  
1.08627369526  
1.10652380952
```

Geschwindigkeitsmessung

```
>fv1(h), ...  
>fv2(h), ...  
>fv3(h)
```

```
95.2593857038  
94.2445842582  
95.1189731211
```

Geschwindigkeitsfehler

```
>aafv1(h), ...  
>aafv2(h), ...  
>aafv3(h)
```

```
-4.74061429622  
-5.75541574179  
-4.88102687887
```